通訊系統實習(二)

主題：Maple 數學應用軟體

目的：學習 Maple 軟體之操作並應用於求解通訊系統所遇到的數學問題

說明：

學通訊的學生最大的障礙莫過於數學的問題了，訊號經過調變、發射、受干擾、接收、解調等一連串的過程，都需要以數學來描述它，甚至是利用數學來求得最後的波形及結果。如果以往沒有把數學學好，包括三角函數、微積分及工程數學等，在學通訊時就會覺得痛苦。幸好拜電腦科技的進步，不需要您用手算，這些複雜的數學運算都可以透過電腦軟體來完成，尤其是利用繪圖，甚至是動畫，的方式來呈現這些數學式子所代表的意義，可以讓你的學習變得有趣一點。

本單元所介紹的是一套非常好用的數學軟體 Maple，它是當今世界上最優秀的幾個數學軟體之一，它以良好的使用環境、強有力的符號計算、高精度的數值計算、靈活的圖形顯示和高效的編程功能，為越來越多的教師、學生和科學研究人員所喜愛，並成為他們進行數學處理的工具。Maple 軟件適用於解決微積分、解析幾何、線性代數、微分方程、計算方法、概率統計等數學分支中的常見計算問題。對於往後通訊原理及電路分析設計的學習都很有幫助。

準備器材：

1. 個人電腦
2. 軟體：Maple

操作進度：

Part I. 基本運算與繪圖

1 **restart:**
相當於將所有宣告的記憶體清除，初學者最好養成習慣，在執行運算之初先執行 **restart:** 的動作。
2 基本運算： **+ - * / sqrt() ^**
3 微積分：**diff() int()**
4 宣告變數及函數：
宣告某個變數 **:=** 。例如 **x:=3; y:=x-6;** 。
如果是宣告某個函數，例如 $x(t)=3t^2+5t-9$，則 Maple 的寫法如下：
[> **x:=t->3*t^2+5*t-9;**
5 Maple 內定的函數及變數
(1) 圓周率 π：**Pi**(要注意 P 是大寫 i 是小寫)
(2) 虛數 $\sqrt{-1}$：**I**
(3) 三角函數：**sin() cos() tan() cot() sec() csc()**
(4) 反三角函數：**arcsin() arcos() arctan()**...
(5) 指數函數：**exp()**
(6) 自然對數函數：**log(x)**或 **ln(x)**
(7) 以 10 為底，或是以 2 為底的對數函數：**log[10](x) log[2](x)**
6 求估測值並以浮點數表示： **evalf()**
7 求估測值並以複數(a+bj)表示： **evalc()**
8 函數定義 x(t), z(x,y)
9 一維繪圖 格式如下：**plot(函數,變數範圍,加選功能 1,加選功能 2,…加選功能 N);**
函數：例如 **x(t)** 或 **G(f)**
變數範圍：例如 **t=-2..2** 或 **f=0..5**
加選功能：包括曲線的顏色(例如 **color=blue**)、線條粗細 (例如 **thickness=2**)、描點數(例如 **numpoints=300**)
實例：
plot(x(t),t=0..2,color=red,thickness=3,numpoints=500);
10 多曲線比較：假如有 M 個相同變數的函數，例如 $x_1(t),x_2(t)\cdots,x_M(t)$，我們想要將這些函數的曲線畫在同一個圖上，Maple 的格式如下：
plot([函數 1,函數 2,…,函數 M],變數範圍,加選功能 1,加選功能 2,…加選功能 N);
11 複數運算
(1) 宣告複數 X=3+5j：**x:=3+5*I;**
(2) 求複數 X 的實部 A 及虛部 B： **A:=Re(X); B:=Im(X)**
(3) 求複數 X 的振幅 R： **R:=abs(X);**
(4) 求複數 X 的相位角 θ(以徑度表示)：**theta:=argument(X);**
(5) 複數算結果 Z 最後以 A+Bj 的形式表示，其中 A 為實部，B 為虛部 **evalc(Z);**

綜 合 練 習

《練習 1》

求以下變數的值，答案請以浮點數的形式表示

(1) $x = \log(18.5)$

(2) $y = \log_2\left[1 + \sin(0.15) \cdot e^{-2}\right]$

(3) $z = \ln\left(4 + 3.5 \cdot e^{0.5\pi}\right) \cdot \sin\left(\frac{\pi}{3}\right)$

(4) $w = \int_3^5 \left(2t^2 + 5\right) dt$

(5) $u = \frac{d}{dt}\left[e^t \sin(30t)\right]$

```
> restart:
> x:=log[10](18.5);
```

$$x := 1.267171728$$

```
> y:=log[2](1+sin(0.15)*exp(-2));
```

$$y := \frac{\ln(1 + 0.1494381325\, \mathbf{e}^{(-2)})}{\ln(2)}$$

```
> evalf(y);
```

$$0.02888630090$$

```
> z:=ln(4+3.5*exp(0.5*Pi))*sin(Pi/3);
```

$$z := \frac{1}{2} \ln(4 + 3.5\, \mathbf{e}^{(0.5\, \pi)})\, \sqrt{3}$$

```
> evalf(z);
```

$$2.629871864$$

```
> w:=int((2*t^2+5),t=3..5);
```

$$w := \frac{226}{3}$$

```
> u:=diff(exp(t)*sin(30*t),t);
```

$$u := \mathbf{e}^t \sin(30\, t) + 30\, \mathbf{e}^t \cos(30\, t)$$

《練習 2》

畫出下列波形

(1) $x(t) = 3\cos(20\pi t) - 2\sin(12\pi t)$

(2) $y(t) = [3 + 2\cos(10t)] \cdot \cos(25\pi t - \pi/6)$

(3) $z(t) = \cos\left(40\pi t - 18\int_0^t 3\sin(5\pi x)\,dx\right)$

(4) $w(t) = 6e^{-2t} - 3t^2 e^{-t}$

```
> restart:
> x:=t->3*cos(20*Pi*t)-2*sin(12*Pi*t);
```

$$x := t \to 3\cos(20\,\pi\,t) - 2\sin(12\,\pi\,t)$$

```
> y:=t->(3+2*cos(10*t))*cos(25*Pi*t-Pi/6);
```

$$y := t \to (3 + 2\cos(10\,t))\cos\left(25\,\pi\,t - \frac{\pi}{6}\right)$$

```
> z:=t->cos(40*Pi*t-32*int(2*cos(5*Pi*x),x=0..t));
```

$$z := t \to \cos\left(40\,\pi\,t - 32\int_0^t 2\cos(5\,\pi\,x)\,dx\right)$$

```
> w:=t->6*exp(-2*t)-3*t^2*exp(-t);
```

$$w := t \to 6\,\mathbf{e}^{(-2\,t)} - 3\,t^2\,\mathbf{e}^{(-t)}$$

```
> plot(x(t),t=0..1);
```




```
> plot(y(t),t=0..3);
```



```
> plot(z(t),t=0..4,numpoints=500);
```



```
> plot(w(t),t=0..15);
```



《練習 3》
畫出 $x(t)=3\cos(40\pi t-\pi/3)$ 的波形並指出 $t=0.1$ 時的波形高度。

```
> restart:
> x:=t->3*cos(40*Pi*t-Pi/3);
```

$$x := t \rightarrow 3\cos\left(40\,\pi\,t-\frac{\pi}{3}\right)$$

```
> plot(x(t),t=0..0.2);
```




```
> evalf(x(0.1));
```

$$1.500000018$$

《練習 4》
畫出 $x(t)=\left[3\sin(10\pi t)+5\cos(6\pi t)\right]\cdot e^{-2t}$ 在 t=0 到 2 之間的波形，並指出 t=0.1 時的波形高度。

```
> restart:
> x:=t->(3*sin(10*Pi*t)+5*cos(6*Pi*t))*exp(-2*t);
```

$$x := t \rightarrow (3\sin(10\,\pi\,t)+5\cos(6\,\pi\,t))\,\mathbf{e}^{(-2\,t)}$$

```
> plot(x(t),t=0..2);
```




```
> evalf(x(0.1));
```

$$-1.265008582$$

《練習 5》

將以下所列函數的波形畫在同一張圖上，並以不同顏色顯示

$x(t)=3\cos(3\pi t)$

$y(t)=3\cos(3\pi t-\pi/4)$

```
> restart:
> x:=t->3*cos(3*Pi*t);
```

$$x := t \to 3\cos(3\,\pi\,t)$$

```
> y:=t->3*cos(3*Pi*t-Pi/4);
```

$$y := t \to 3\cos\left(3\,\pi\,t-\frac{\pi}{4}\right)$$

```
> plot([x(t),y(t)],t=-1..1,color=[red,blue]);
```



《練習 6》

假設 A=5+8j，B=7-9j，C=12-4j，求以下複數的實部 A，虛部 B，振幅 R，以及相角 θ (請將徑度轉換成角度)

(1)　$Z_1=(A+B)\cdot C$

(2)　$Z_2=(A-B)/C$

```
> restart:
> A:=5+8*I; B:=7-9*I; C:=12-4*I;
```

$$A := 5+8\,I$$
$$B := 7-9\,I$$
$$C := 12-4\,I$$

```
> Z1:=(A+B)*C;
```

$$Z1 := 140-60\,I$$

```
> abs(Z1); argument(Z1);
```

$$20\sqrt{58}$$
$$-\arctan\left(\frac{3}{7}\right)$$

```
> evalf(argument(Z1)*180/Pi);
```

$$-23.19859051$$

```
> Z2:=(A-B)/C;
```

$$Z2 := \frac{-23}{40}+\frac{49}{40}\,I$$

```
> abs(Z2); argument(Z2);
```

$$\frac{\sqrt{2930}}{40}$$

$$-\arctan\left(\frac{49}{23}\right)+\pi$$

```
> evalf(argument(Z2)*180/Pi);
```

$$115.1447856$$

練　習

一、求以下變數的值，答案請以浮點數的形式表示

(1) $x=\log(92.3)$

(2) $y=\log_2\left[2+\cos(0.12\pi)\cdot e^{-4}\right]$

(3) $z=\ln\left(4+\frac{2}{5}\cdot e^{\frac{3}{5}\pi}\right)\cdot\sin\left(\frac{\pi}{3}\right)$

(4) $w=\int_1^2\left(3t^5+12\right)e^{-3t^3}dt$

(5) $u=\frac{d}{dt}\left[2t^{-2}e^t\cos(0.3t)\right]$

二、畫出下列波形

(1) $x(t)=\left[3\cos(12\pi t-2\pi/5)-2\sin(7\pi t+9\pi/4)\right]e^{-t}$

(2) $y(t)=\left[1+2\sin(8t)\right]\cdot\cos(25t-\pi/6)$

(3) $z(t)=3\cos\left(25\pi t-30\int_0^t\left[2\sin(5\pi x)+3\cos(7\pi x)\right]dx\right)$

(4) $w(t)=6e^{-t}+t^2e^{-2t}$

三、將以下所列函數的波形畫在同一張圖上，並以不同顏色顯示

$x(t)=3\sin(3\pi t)$

$y(t)=3\sin(3\pi t+\pi/5)$

$z(t)=3\sin(3\pi t-\pi/5)$

四、假設 A=4-6j，B=5-9j，C=8+3j，求以下複數的實部 A，虛部 B，振幅 R，以及相角 θ（請將徑度轉換成角度）

(1) $Z_1=(A+B)\cdot C$

(2) $Z_2=(A-B)/C$

Part II. 以 Maple 求對稱及非對稱方波的傅利葉轉換

對稱方波的傅利葉轉換

```
> restart:
```

Step 1: 宣告 g(t)爲高度 A 寬度 tau 的對稱方波，並畫出波形

```
> g:=t->A*(Heaviside(t+tau/2)-Heaviside(t-tau/2));
```

$$g := t \rightarrow A\left(\mathrm{Heaviside}\left(t+\frac{\tau}{2}\right)-\mathrm{Heaviside}\left(t-\frac{\tau}{2}\right)\right)$$

```
> g(-tau/2):=A/2; g(tau/2):=A/2;
```

#因爲 **t=-tau/2** 及 **t=tau/2** 處爲奇點，因此必須給定一個實際數値

$$\mathrm{g}\left(-\frac{\tau}{2}\right) := \frac{A}{2}$$

$$\mathrm{g}\left(\frac{\tau}{2}\right) := \frac{A}{2}$$

```
> A:=3;tau:=4;
```

$$A := 3$$

$$\tau := 4$$

```
> plot(g(t),t=-5..5);
```



Step 2: 以傅利葉轉換求 g(t)的頻譜 G(f)並畫出 g(t)的振幅頻譜及相位頻譜

```
> with(inttrans):
> G:=f->fourier(g(t),t,2*Pi*f);
```

$$G := f \rightarrow \mathrm{fourier}(\,\mathrm{g}(t),\, t,\, 2\,\pi\, f)$$

```
> plot(G(f),f=-3..3);
```



g(t)的振幅頻譜爲 G(f)的絕對值 |G(f)|

```
> plot(abs(G(f)),f=-3..3);
```



#定義 Phi(f)= arctan(G(f)的虛部/G(f)的實部)，畫出 Phi(f)在-Pi..Pi 範圍的圖即為 G(f)的相位頻譜

```
> Phi:=f->arctan(Im(G(f))/Re(G(f)));
```

$$\Phi := f \rightarrow \arctan\left(\frac{\Im(\mathrm{G}(f))}{\Re(\mathrm{G}(f))}\right)$$

由於對稱的波形沒有虛數，也就是 Im(G(f))=0，因此對所有頻率 f 而言，相位均為 0，(Phi(f)=0)

```
> plot(Phi(f),f=-Pi..Pi,thickness=3);
```



進階：以下利用動畫表現出 g(t)脈波寬度改變與其相對應的頻譜 G(f)的變化關係

```
> restart:
> g:=t->A*(Heaviside(t+tau/2)-Heaviside(t-tau/2));
```

$$g := t \rightarrow A\left(\mathrm{Heaviside}\left(t+\frac{\tau}{2}\right)-\mathrm{Heaviside}\left(t-\frac{\tau}{2}\right)\right)$$

```
> g(t-tau/2):=A/2; g(t+tau/2):=A/2;
```

$$\mathrm{g}\left(t-\frac{\tau}{2}\right) := \frac{A}{2}$$

$$\mathrm{g}\left(t+\frac{\tau}{2}\right) := \frac{A}{2}$$

```
> with(inttrans):
> G:=f->fourier(g(t),t,2*Pi*f);
```

$$G := f \rightarrow \mathrm{fourier}(\mathrm{g}(t), t, 2\,\pi\,f)$$

```
> A:=3;
```

$$A := 3$$

```
> with(plots):
Warning, the name changecoords has been redefined

> U:=array(1..2):
```

```
>
U[1]:=animate(g(t),t=-10..10,tau=0.1..10,numpoints=500,title="g(t)
"):
>
U[2]:=animate(G(f),f=-3..3,tau=0.1..10,numpoints=300,title="G(f)")
:
> display(U);
```





(註：本圖為動畫)

非對稱方波的傅利葉轉換

```
> restart:
```

Step 1: 宣告 g(t)爲高度 A 寬度 tau 且往前偏移 t0 的方波，並畫出波形

```
> g:=t->A*(Heaviside(t+tau/2-t0)-Heaviside(t-tau/2-t0));
```

$$g := t \to A\left(\text{Heaviside}\left(t+\frac{\tau}{2}-t0\right)-\text{Heaviside}\left(t-\frac{\tau}{2}-t0\right)\right)$$

```
> g(-tau/2+t0):=A/2; g(tau/2+t0):=A/2; #¦]¬°t=-tau/2 ¤Î t=tau/2
³B¬°©_ÂI¡A¦]¦¹¥²¶·µ¹©w¤@-Ó¹ê»Ú¼Æ-È
```

$$g\left(-\frac{\tau}{2}+t0\right) := \frac{A}{2}$$

$$g\left(\frac{\tau}{2}+t0\right) := \frac{A}{2}$$

```
> A:=3;tau:=4;t0:=1;
```

$$A := 3$$

$$\tau := 4$$

$$t0 := 1$$

```
> plot(g(t),t=-5..5);
```



Step 2: 以傅利葉轉換求 g(t)的頻譜 G(f)並畫出 g(t)的振幅頻譜及相位頻譜

```
> with(inttrans):
> G:=f->fourier(g(t),t,2*Pi*f);
```

$$G := f \to \text{fourier}(g(t), t, 2\pi f)$$

#別想直接畫出 G(f)的圖，因爲 G(f)是複數，只能畫出 G(f)的振幅及相位 ì

```
> plot(G(f),f=-3..3);
```



g(t)的振幅頻譜爲 G(f)的絕對值 |G(f)| ，事實上它與對稱方波的振幅頻譜完全相同

```
> plot(abs(G(f)),f=-3..3);
```



#定義 Phi(f)= arctan(G(f)的虛部/G(f)的實部)，畫出 Phi(f)在-Pi..Pi 範圍的圖即為 G(f)的相位頻譜

```
> Phi:=f->arctan(Im(G(f))/Re(G(f)));
```

$$\Phi := f \to \arctan\left(\frac{\Im(\mathrm{G}(f))}{\Re(\mathrm{G}(f))}\right)$$

```
> plot(Phi(f),f=-Pi..Pi);
```



進階：以下是利用動畫顯示出方波位移時，振幅頻譜與相位頻譜的變化情形，結論是振幅頻譜不會因為時間位移而有所改變

```
> restart:
> g:=(t)->A*(Heaviside(t+tau/2-t0)-Heaviside(t-tau/2-t0));
```

$$g := t \to A\left(\mathrm{Heaviside}\left(t + \frac{\tau}{2} - t0\right) - \mathrm{Heaviside}\left(t - \frac{\tau}{2} - t0\right)\right)$$

```
> g(-tau/2+t0):=A/2; g(tau/2+t0):=A/2;
```

$$\mathrm{g}\left(-\frac{\tau}{2} + t0\right) := \frac{A}{2}$$

$$\mathrm{g}\left(\frac{\tau}{2} + t0\right) := \frac{A}{2}$$

```
> with(inttrans):
> G:=f->fourier(g(t),t,2*Pi*f);
```

$$G := f \to \mathrm{fourier}(\mathrm{g}(t), t, 2\,\pi\,f)$$

```
> Phi:=f->arctan(Im(G(f,t0))/Re(G(f,t0)));
```

$$\Phi := f \to \arctan\left(\frac{\Im(\mathrm{G}(f, t0))}{\Re(\mathrm{G}(f, t0))}\right)$$

```
> A:=3; tau:=2;
```

$$A := 3$$

$$\tau := 2$$

```
> with(plots):
Warning, the name changecoords has been redefined

> U:=array(1..3):
>
U[1]:=animate(g(t),t=-10..10,t0=-6..6,numpoints=500,frames=13,titl
e="Time Domain Waveform"):
>
U[2]:=animate(abs(G(f)),f=-Pi..Pi,t0=-6..6,numpoints=300,frames=13
,title="Amplitude Spectrum"):
>
U[3]:=animate(Phi(f),f=-Pi..Pi,t0=-6..6,numpoints=300,frames=13,ti
tle="Phase Spectrum"):
> display(U);
```



（註：本圖爲動畫）

```
>
```

Part III. 以 Maple 求週期函數的傅利葉級數

1. 題目：以傅利葉級數表示以下圖所示的週期函數 $g_1(t)$



2. 理論推導

$$g_1(t)=a_0+\sum_{n=1}^{\infty}\left[a_n\cos\frac{2\pi nt}{T_0}+b_n\sin\frac{2\pi nt}{T_0}\right],\ \text{其中：}$$

$$a_0=\frac{1}{T_0}\int_0^T g_1(t)dt=\frac{1}{3}\left(\int_0^1 3dt+\int_1^2 1\cdot dt\right)=\frac{4}{3}$$

$$a_n=\frac{2}{T_0}\int_0^T g_1(t)\cos\frac{2\pi nt}{T_0}dt$$

$$=\frac{2}{3}\left(\int_0^1 3\cdot\cos\frac{2\pi nt}{3}dt+\int_1^2 1\cdot\cos\frac{2\pi nt}{3}dt\right)=\frac{1}{n\pi}\left(2\cdot\sin\frac{2n\pi}{3}+\sin\frac{4n\pi}{3}\right)$$

$$b_n=\frac{2}{T_0}\int_0^T g_1(t)\sin\frac{2\pi nt}{T_0}dt$$

$$=\frac{2}{3}\left(\int_0^1 3\cdot\sin\frac{2\pi nt}{3}dt+\int_1^2 1\cdot\sin\frac{2\pi nt}{3}dt\right)=\frac{-1}{\pi n}\left(2\cdot\cos\frac{2n\pi}{3}+\cos\frac{4n\pi}{3}-3\right)$$

3. Maple 範例：求解推導 a_0、a_n 及 b_n 過程中的積分部分(以紅色標示)可藉由 Maple 來完成，再將此三個所得參數代入原式即可求得 $g_1(t)$ 的傅利葉級數。

```
> restart:
> #定義 Fourier series 的通式
>g:=t->a0+Sum('a(n)*cos(2*Pi*n*t/T0)+b(n)*sin(2*Pi*n*t/T0)','n'=1..N);
```

$$g:=t\to a0+\left(\sum_{'n'=1}^{N}\text{'}\mathrm{a}(n)\cos\left(\frac{2\,\pi\,n\,t}{T0}\right)+\mathrm{b}(n)\sin\left(\frac{2\,\pi\,n\,t}{T0}\right)\text{'}\right)$$

```
> #列出 a0,an 及 bn 的運算式，如果將 Int 改爲 int，maple 會直接顯示出 a0 的答案(a0=4/T0)
> a0:=1/T0*(Int(3,t=0..1)+Int(1,t=1..2));
```

$$a0:=\frac{\int_0^1 3\,dt+\int_1^2 1\,dt}{T0}$$

```
>
a:=n->2/T0*(int(3*cos(2*Pi*n*t/T0),t=0..1)+int(1*cos(2*Pi*n*t/T0
),t=1..2));
```

$$a := n \to \frac{2\left(\int_0^1 3\cos\left(\frac{2\pi n t}{T0}\right)dt + \int_1^2 \cos\left(\frac{2\pi n t}{T0}\right)dt\right)}{T0}$$

```
>
b:=n->2/T0*(int(3*sin(2*Pi*n*t/T0),t=0..1)+int(1*sin(2*Pi*n*t/T0
),t=1..2));
```

$$b := n \to \frac{2\left(\int_0^1 3\sin\left(\frac{2\pi n t}{T0}\right)dt + \int_1^2 \sin\left(\frac{2\pi n t}{T0}\right)dt\right)}{T0}$$

> #給定 **g(t)**的週期 **T0** 以及累加的項數 **N(N** 的理論值爲無窮大**)**

```
> T0:=3;N:=10;
```

$$T0 := 3$$
$$N := 10$$

> #如果需要的話可列出 **an** 及 **bn** 的計算結果**,**加上 **simplify** 指令可將結果簡化

```
> a(n);
```

$$\frac{3\sin\left(\frac{2\pi n}{3}\right)}{\pi n} + \frac{\sin\left(\frac{4\pi n}{3}\right) - \sin\left(\frac{2\pi n}{3}\right)}{\pi n}$$

```
> simplify(a(n));
```

$$\frac{2\sin\left(\frac{2\pi n}{3}\right) + \sin\left(\frac{4\pi n}{3}\right)}{\pi n}$$

```
> simplify(b(n));
```

$$-\frac{2\cos\left(\frac{2\pi n}{3}\right) - 3 + \cos\left(\frac{4\pi n}{3}\right)}{\pi n}$$

> #畫出 **g(t)**的波形**,**顯然 **N=10** 的精度還不夠，還有一項缺點就是計算速度慢

```
> plot(g(t),t=-5..5,thickness=3);
```



> #爲了計算快速，重新定義所有函數，避免使用到積分**(**也就是將已知的 **a0,an,bn** 直接代入**)**

```
> restart:
>
```

```
g:=t->a0+Sum('a(n)*cos(2*Pi*n*t/T0)+b(n)*sin(2*Pi*n*t/T0)','n'=1..N);
```

$$g := t \to a0 + \left(\sum_{'n'=1}^{N} \text{'a}(n)\cos\left(\frac{2\,\pi\,n\,t}{T0}\right) + \text{b}(n)\sin\left(\frac{2\,\pi\,n\,t}{T0}\right)\text{'} \right)$$

```
> a0:=4/T0;
```

$$a0 := \frac{4}{T0}$$

```
> a:=n->(2*sin(2/3*Pi*n)+sin(4/3*Pi*n))/Pi/n;
```

$$a := n \to \frac{2\sin\left(\frac{2}{3}\pi\,n\right) + \sin\left(\frac{4}{3}\pi\,n\right)}{\pi\,n}$$

```
> b:=n->-(2*cos(2/3*Pi*n)-3+cos(4/3*Pi*n))/Pi/n;
```

$$b := n \to -\frac{2\cos\left(\frac{2}{3}\pi\,n\right) - 3 + \cos\left(\frac{4}{3}\pi\,n\right)}{\pi\,n}$$

> #將 **N** 改為 **100**，所的結果與原始波形就很接近了，且即使 **N** 很大，以下的計算速度仍很快

```
> T0:=3;N:=100;  #注意 T0 仍為 3
```

$$N := 100$$

```
> plot(g(t),t=-5..5,thickness=2,numpoints=300);
```



>

Part IV. 以 Maple 求解微分方程

1. 假設微分方程為：

$$\frac{d^2v}{dt^2}+6\frac{dv}{dt}+5v=0$$

2. 假設 initial condidtion $v(0)=4,\ \dfrac{dv(0)}{dt}=-24$

3. 求解微分方程的通解及特解並畫出 $v(t)$ 的曲線圖

```
> restart:
> del:=diff(v(t),t$2)+6*diff(v(t),t)+5*v(t);
```

$$del := \left(\frac{d^2}{dt^2}\mathrm{v}(t)\right)+6\left(\frac{d}{dt}\mathrm{v}(t)\right)+5\,\mathrm{v}(t)$$

```
> dsolve(del,v(t));
```

$$\mathrm{v}(t)=\_C1\,\mathbf{e}^{(-t)}+\_C2\,\mathbf{e}^{(-5\,t)}$$

```
> ini:=v(0)=4,D(v)(0)=-24;
```

$$ini := \mathrm{v}(0)=4,\ \mathrm{D}(v)(0)=\text{-}24$$

```
> dsolve({del,ini},{v(t)});
```

$$\mathrm{v}(t)=-\mathbf{e}^{(-t)}+5\,\mathbf{e}^{(-5\,t)}$$

```
> with(DEtools):
> DEplot(del,v(t),0..5,[[ini]],stepsize=0.1);
```




```
>
```

Part V. 以 Maple 模擬 DSB 通訊系統

```
> restart:
```

調變

```
> #定義欲傳送信號m(t)及載波c(t)的波形
>m:=t->0.6*cos(2*Pi*fm1*t)-5.7*cos(2*Pi*fm2*t)+2.5*cos(2*Pi*fm3*t)
;
```

$$m := t \to 0.6\cos(2\pi\, fm1\, t) - 5.7\cos(2\pi\, fm2\, t) + 2.5\cos(2\pi\, fm3\, t)$$

```
> c:=t->cos(2*Pi*fc*t);
```

$$c := t \to \cos(2\pi\, fc\, t)$$

```
> fm1:=300;fm2:=700;fm3:=1300;fc:=10000;
```

$$fm1 := 300$$
$$fm2 := 700$$
$$fm3 := 1300$$
$$fc := 10000$$

```
> s:=t->m(t)*c(t);
```

$$s := t \to \mathrm{m}(t)\,\mathrm{c}(t)$$

```
> plot([m(t),s(t)],t=0..0.006,color=[red,green],numpoints=300);
```



解調

```
> #在無雜訊情況下，假設本地振盪器與載波的相位差為 theta=60 度
> x1:=t->s(t)*cos(2*Pi*fc*t+theta);theta:=60*Pi/180;
```

$$x1 := t \to \mathrm{s}(t)\cos(2\pi\, fc\, t + \theta)$$

$$\theta := \frac{\pi}{3}$$

```
> plot([m(t),x1(t)],t=0..0.006,color=[red,green],numpoints=300);
```




```
> #經過低通濾波器將高頻信號濾除即為解調信號
> sinc:=t->sin(Pi*t)/(Pi*t);
```

$$sinc := t \to \frac{\sin(\pi\, t)}{\pi\, t}$$

```
> F:=t->2*B*sinc(2*B*t); B:=1500; #設定濾波器頻寬為1.5kHz(比m(t)頻寬大即可)
```

$$F := t \to 2\, B\, \mathrm{sinc}(2\, B\, t)$$
$$B := 1500$$

```
> yD:=t->int(x1(t-tau)*F(tau),tau=-infinity..infinity);
```

$$yD := t \to \int_{-\infty}^{\infty} \mathrm{x1}(t-\tau)\, \mathrm{F}(\tau)\, d\tau$$

```
> plot([m(t),yD(t)],t=0..0.006,color=[red,blue]);
```




```
> #如果設定本地振盪器與載波同步(theta=0)，則解調信號沒有任何衰減(亦即yD(t)=1/2*m(t))
> theta:=0;
```

$$\theta := 0$$

```
> plot([m(t),yD(t)],t=0..0.006,color=[red,blue]);
```




```
>
```

Part VI. 以 Maple 模擬 FM 通訊系統

FM 調變及解調模擬：假設基頻信號 m(t)= .4*sin(10*Pi*t)+.7*cos(6*Pi*t);
若以 FM 發射機傳送，振幅為 3，頻率靈敏度為 kf=25Hz/V，中心頻率為 50Hz
(1)畫出 FM 信號的調變及解調過程
(2)指出該 FM 信號的調變指數及頻寬

```
> restart:
```

STEP 1: 定義基頻信號數學式

```
> m:=t->0.4*sin(2*Pi*fm1*t)+0.7*cos(2*Pi*fm2*t);
```

$$m := t \to 0.4 \sin(2\,\pi\, fm1\; t) + 0.7 \cos(2\,\pi\, fm2\; t)$$

STEP2: 調變

基頻信號經 FM 調變後的 FM 信號數學式 s_FM(t)宣告

```
> s_FM:=t->A*cos(2*Pi*fc*t+2*Pi*kf*int(m(alpha),alpha=0..t));
```

$$s\_FM := t \to A \cos\left(2\,\pi\, fc\; t + 2\,\pi\, kf \int_0^t \mathrm{m}(\alpha)\, d\alpha\right)$$

```
> #µ¹©w°Ñ¼Æ-È
> fm1:=5;fm2:=3; fc:=50; kf:=25;A:=3;
```

$$fm1 := 5$$
$$fm2 := 3$$
$$fc := 50$$
$$kf := 25$$
$$A := 3$$

顯示基頻信號 m(t) 波形

```
> plot(m(t),t=0..1);
```



基頻信號與 FM 信號的比較，注意基頻信號振幅越大，FM 信號瞬間頻率則越高

```
> plot([m(t),s_FM(t)],t=0..1);
```




```
> kf:=10;
```

$$kf := 10$$

```
> plot([m(t),s_FM(t)],t=0..1,numpoints=300);
```



STEP 3:解調

#解調 FM 信號的第一個步驟是經過微分器，微分輸出的波封爲

r1(t)=A(2*Pi*fc+2*pi*kf*m(t));

```
>
> r1:=t->diff(s_FM(t),t);
```

$$r1 := t \rightarrow \mathrm{diff}(\mathrm{s\_FM}(t), t)$$

#反推求得波峰爲 fc/fd+m(t)的數學式，定義爲 r2(t)

```
> r2:=t->(r1(t)/A)/(2*Pi*kf);
```

$$r2 := t \rightarrow \frac{1}{2} \frac{\mathrm{r1}(t)}{A \pi kf}$$

#畫出 r2(t)與 m(t)之比較

```
> plot([m(t),r2(t)],t=0..1);
```



#一般而言，若 r2(t)經過波峰檢測器後再減去 fc/fd，其結果應為 m(t)

```
> plot([m(t),r2(t)-fc/kf],t=0..1);
```



#以上模擬均省略濾波器的動作
#本系統的調變指數為

```
> beta:=Am*kf/fm;
```

$$\beta := \frac{25\ Am}{fm}$$

帶入 Am=1.1, fm=5 則調變指數為 5.5

```
> Am:=1.1;fm:=5;
```

$$Am := 1.1$$
$$fm := 5$$

```
> evalf(beta);
```

$$5.500000000$$

根據卡爾森規則，FM 信號頻寬為 65Hz

```
> B:=2*(1+beta)*fm;
```

$$B := 65.00000000$$